TTM1 - A FORTRAN IMPLEMENTATION OF THE TELLEGEN THEOREM METHOD TO POWER SYSTEM SIMULATION AND DESIGN

J.W. Bandler, M.A. El-Kady and J. Wojciechowski

SOS-82-12-L2

July 1983

TTM1 - A FORTRAN IMPLEMENTATION OF THE TELLEGEN THEOREM METHOD TO POWER SYSTEM SIMULATION AND DESIGN

J.W. Bandler, M.A. El-Kady and J. Wojciechowski

Abstract

This report contains a listing of the computer package TTM1 described in [1] for reduced gradient evaluation and load flow solution of power networks. The package has been developed for the CDC 170/730 system with the NOS 1.4 level 552 operating system and the Fortran Extended (FTN) version 4.8 compiler. The listing contains a total of 1916 lines (including 683 comments) constituting twenty-five subroutines. The listing does not include the Harwell package MA28 [2] for solving sparse linear equations.

This work was supported by the Natural Sciences and Engineering Research Council of Canada under Grant G0647.

The authors are with the Simulation Optimization Systems Research Laboratory and the Department of Electrical and Computer Engineering, McMaster University, Hamilton, Canada L8S 4L7.

M.A. El-Kady is also with Ontario Hydro, Toronto, Canada.

J. Wojciechowski is on leave from the Institute of Electronics Fundamentsals, Technical University of Warsaw, Warsaw, Poland.

1 INTRODUCTION

This document contains a Fortran listing of all alphabetically ordered subroutines of the TTM1 package. Essentially, the TTM1 package is for reduced gradients evaluation and load flow solution of power networks, though it also contains data handling routines. The user's manual of the TTM1 package together with illustrative examples is found in [1].

The TTM1 package has been developed for the CDC 170/730 system with the NOS 1.4 level 552 operating system and the Fortran Extended (FTN) version 4.8 compiler. The package is available at McMaster University in the form of a library of binary relocatable subroutines. The library is in the group indirect file LIBTTM1 accessible under the charge RJWBAND. The package calls subroutines MA28A, MA28B and MA28C of the Harwell Subroutine Library (Harwell package MA28) for solving sparse linear equations; the MA28 package must thus also be available when TTM1 is used. This document does not include the MA28 package: information concerning this package is found in [2].

The TTM1 package contains 1916 lines of which 683 are comments. It has been modularized into 25 subroutines. The list of all subroutines with short statistics is given in Table 1.

TABLE 1

LIST OF SUBROUTINES OF THE TTM1 PACKAGE

	Subroutine	Number of lines (source text)	Number of words (compiled code)	Description (page of [1])	Listing (page)
1	CURR	20	50	90	5
2	DCVARF	36	64	92	6
3	DERIV	91	604	94	7
4	EXCT	78	273	101	9
5	FORMDTF	39	422	103	11
6	FORMPR	59	441	106	12
7	FORMTA	87	512	110	13
8	FORMTAD	80	413	113	15
9	FORMTD	89	472	115	17
10	FORMTE	99	624	117	19
11	FORMTM	100	606	120	21
12	FORMU	45	203	123	23
13	FORMYT	63	315	126	24
14	LFLFD1M	93	274	130	25
15	LFLFDAM	192	1473	134	27
16	LFLFDBM	42	203	141	30
17	LFLFDCM	46	213	144	31
18	LFTTM	157	721	147	32
19	MISM	47	253	152	35
20	PQ	31	210	154	36
21	PTELT	19	34	156	37
22	RDAT	84	720	158	38
23	READDT	76	564	162	40
24	SENSIT	77	635	167	42
25	STEP	166	1270	173	44

2 REFERENCES

[1] J.W. Bandler, M.A. El-Kady and J. Wojciechowski, "TTM1 - A Fortran implementation of the Tellegen theorem method to power system simulation and design", Department of Electrical and Computer Engineering, McMaster University, Hamilton, Canada, Report SOS-82-12-U2, 1983.

[2] I.S. Duff, "MA28 - A set of Fortran subroutines for sparse unsymmetric linear equations", Computer Science and Systems Division, AERE Harwell, Oxfordshire, England, Report R.8730, 1980.

3 LISTING OF THE TTM1 PACKAGE

```
C
C
C
      COMPLEX FUNCTIONCURR(X,YT,JRYT,ICYT,MB)
C
C     12.05.1982.
C
C     FUNCTION  SUBPROGRAM  CURR  CALCULATES  THE VALUE OF THE  CURRENT
C     INJECTED INTO THE MBTH BUS FOR THE GIVEN VECTOR X OF BUS VOLTAGES
C
      INTEGER JRYT(1),ICYT(1)
      COMPLEX X(1),YT(1)
      K1=JRYT(MB)
      K2=JRYT(MB+1)-1
      CURR=(0.,0.)
      DO 10 I=K1,K2
      CURR=CURR+YT(I)*X(ICYT(I))
   10 CONTINUE
      RETURN
      END
```

```
C
C
C
      SUBROUTINE DCVARF (CCVF,ICVF,BTYP,N)
C
C     19.05.1982.
C
C     SUBROUTINE DCVARF DECLARES CONTROL VARIABLES FOR THE LOAD
C     FLOW PROBLEM
C
      INTEGER BTYP(1),CCVF(1),ICVF(1)
      DO 20 I=1,N
      J=2*I-1
      IF (BTYP(I).EQ.1) GO TO 10
C
C     SETTING UP VALUES OF ELEMENTS OF CCVF, ICVF CORRESPONDING
C     TO A LOAD BUS
C
      CCVF(J)=7
      ICVF(J)=I
      J=J+1
      CCVF(J)=8
      ICVF(J)=I
      GO TO 20
C
C     SETTING UP VALUES OF ELEMENTS OF CCVF, ICVF CORRESPONDING
C     TO A GENERATOR BUS
C
   10 CCVF(J)=7
      ICVF(J)=I
      J=J+1
      CCVF(J)=9
      ICVF(J)=I
   20 CONTINUE
      RETURN
      END
```

```
C
C
C
      SUBROUTINE DERIV (LBINP,LBOUT,YT,JRYT,ICYT,V,VT,CCV,ICV,PDR,SENS,N
     1CV,IWRITE)
C
C     19.05.1982.
C
C     SUBROUTINE  DERIV  CALCULATES  VECTOR  SENS OF  SENSITIVITIES
C     OF A REAL  FUNCTION  OF THE  POWER  SYSTEM  STATE AND CONTROL
C     VARIABLES  W.R.T. CONTROL VARIABLES SPECIFIED BY VECTORS CCV,
C     ICV WHEN VECTOR VT OF THE SOLUTION OF ADJOINT SYSTEM IS GIVEN
C
      INTEGER LBINP(1),LBOUT(1),JRYT(1),ICYT(1),CCV(1),ICV(1)
      REAL SENS(1),RLB(11)
      COMPLEX V(1),VT(1),YT(1),PDR(1),CS,VA,VAT,CURR
      DATA RLB/"LINPG(","LINPB(","    LG(","    LB(","LOUTG(","LOUTB(","
     1  P(","    Q(","MOD V(","ARG V(","BSTL("/
      I=0
      DO 100 J=1,NCV
      L=CCV(J)
      M=ICV(J)
C
C           LINPG LINPB LG LB LOUTG LOUTB P   Q  MODV ARGV BSTL
      GO TO (10,10,20,20,30,30,60,70,80,90,50), L
C
C     SENSITIVITY W.R.T.  INPUT SHUNT ADMITTANCE
C
   10 K1=LBINP(M)
      VA=V(K1)
      I=I+1
      VAT=VT(K1)+PDR(I)
      GO TO 40
C
C     SENSITIVITY W.R.T.  TRANSMISSION LINE ADMITTANCE
C
   20 K1=LBINP(M)
      K2=LBOUT(M)
      VA=V(K1)-V(K2)
      I=I+1
      VAT=VT(K1)-VT(K2)+PDR(I)
      GO TO 40
C
C     SENSITIVITY W.R.T.  OUTPUT SHUNT ADMITTANCE
C
   30 K1=LBOUT(M)
      VA=V(K1)
      I=I+1
      VAT=VT(K1)+PDR(I)
   40 CS=VA*VAT
      L1=MOD(L,2)
      IF (L1.EQ.1) SENS(J)=-REAL(CS)
      IF (L1.EQ.0) SENS(J)=AIMAG(CS)
      GO TO 100
C
C     SENSITIVITY W.R.T.  BUS STATIC LOAD
C
   50 I=I+1
      SENS(J)=AIMAG(V(M)*(VT(M)+PDR(I)))
      GO TO 100
C
C     SENSITIVITY W.R.T.  BUS ACTIVE POWER
C
   60 SENS(J)=REAL(VT(M)/CONJG(V(M)))
      GO TO 100
```

```
C
C     SENSITIVITY W.R.T. BUS REACTIVE POWER
C
   70 SENS(J)=AIMAG(VT(M)/CONJG(V(M)))
      GO TO 100
C
C     SENSITIVITY W.R.T. MODULUS OF BUS VOLTAGE
C
   80 SENS(J)=-REAL(VT(M)*CURR(V,YT,JRYT,ICYT,M)+V(M)*CURR(VT,YT,JRYT,IC
     1YT,M))/CABS(V(M))
      GO TO 100
C
C     SENSITIVITY W.R.T. ARGUMENT OF BUS VOLTAGE
C
   90 SENS(J)=AIMAG(-VT(M)*CURR(V,YT,JRYT,ICYT,M)+V(M)*CURR(VT,YT,JRYT,I
     1CYT,M))
  100 CONTINUE
      IF (IWRITE.LT.3) GO TO 110
      WRITE (6,120)
      WRITE (6,130) (RLB(CCV(J)),ICV(J),SENS(J),J=1,NCV)
  110 RETURN
  120 FORMAT (//" VECTOR SENS OF SENSITVITIES WITH RESPECT TO CONTROL VA
     1RIABLES"/" VALUE OF SENSITIVITY IS PRECEDED BY THE CONTROL VARIABL
     2E IDENTIFIER"/)
  130 FORMAT ((1X,4(3X,A6,I3,"):",1X,E13.7)))
      END
```

```
C
C
C
      SUBROUTINE EXCT (V,ITYP,I,RHST,NR,IP,IWRITE)
C
C     11.05.1982.
C
C     SUBROUTINE  EXCT  FORMS THE RIGHT  HAND  SIDE VECTOR RHST OF
C     THE  ADJOINT  EQUATIONS  FOR THE ITH STATE  VARIABLE  OF THE
C     POWER FLOW PROBLEM GIVEN IN RECTANGULAR OR POLAR FORMULATION
C
      REAL RHST(1)
      COMPLEX V(1),VA
      L=MOD(I,2)
      K=(I+L)/2
C
C     K IS THE INDEX OF THE BUS DESCRIBED BY THE ITH CONTROL VARIABLE
C
      DO 10 J=1,NR
      RHST(J)=0.
   10 CONTINUE
      J2=2*K
      J1=J2-1
C
C     J1,J2  ARE INDICES OF ELEMENTS OF THE RHST VECTOR WHICH HAVE TO
C     BE MODIFIED
C
      IF (IP.EQ.1) GO TO 30
C
C     RECTANGULAR FORMULATION OF RHST VECTOR`
C
      IF (ITYP.EQ.0) GO TO 20
C
C     SETTING UP VALUES OF ELEMENTS OF THE RHST VECTOR CORRESPONDING
C     TO A GENERATOR
C
      IF (L.EQ.0) RHST(J1)=REAL(V(K))
      IF (L.EQ.1) RHST(J1)=-AIMAG(V(K))
      GO TO 60
C
C     SETTING UP VALUES OF ELEMENTS OF THE RHST VECTOR CORRESPONDING
C     TO A LOAD BUS
C
   20 IF (L.EQ.0) RHST(J2)=1.
      IF (L.EQ.1) RHST(J1)=-1.
      GO TO 60
C
C     POLAR FORMULATION OF RHST VECTOR
C
   30 IF (ITYP.EQ.0) GO TO 40
C
C     SETTING UP VALUES OF ELEMENTS OF THE RHST VECTOR CORRESPONDING
C     TO A GENERATOR BUS
C
      IF (L.EQ.0) RHST(J1)=1.
      GO TO 60
C
C     SETTING UP VALUES OF ELEMENTS OF THE RHST VECTOR CORRESPONDING
C     TO A LOAD BUS
C
   40 IF (L.EQ.1) GO TO 50
      VA=(0.,1.)/V(K)
      RHST(J1)=REAL(VA)
      RHST(J2)=AIMAG(VA)
      GO TO 60
```

```
50 VA=V(K)
   VA=-CABS(VA)/VA
   RHST(J1)=REAL(VA)
   RHST(J2)=AIMAG(VA)
60 IF (IWRITE.LT.4) GO TO 70
   WRITE (6,80) K,L,IP
   WRITE (6,90) (M,RHST(M),M=1,NR)
70 RETURN
80 FORMAT (//" RHS VECTOR RHST OF ADJOINT EQUATIONS. BUS NO.",I3,"
  1L=",I2,"    IP=",I2/" VALUE OF AN ELEMENT IS PRECEDED BY THE EQUATI
  2ON INDEX"/)
90 FORMAT (5(2X,I4,":",E13.7))
   END
```

```
C
C
C
      SUBROUTINE FORMDTF (LBINP,LBOUT,LINPG,LINPB,LR,LX,LOUTG,LOUTB,LTAP
     1,BNR,BTYP,BVMOD,BVARG,BGP,BLP,BLQ,BSTL,HDLN,NB,NTL,OTPT)
C
C     27.04.1982.
C
C     SUBROUTINE FORMDTF CREATES A FORMATTED DATA FILE DESCRIBING
C     THE POWER SYSTEM
C
      INTEGER LBINP(1),LBOUT(1),BNR(1),BTYP(1),OTPT
      REAL LINPG(1),LINPB(1),LR(1),LX(1),LOUTG(1),LOUTB(1),LTAP(1),BVMOD
     1(1),BVARG(1),BGP(1),BLP(1),BLQ(1),BSTL(1),RL(9),RB(8),HDLN(8)
C
C     MATRIX HDLN KEEPS AN IDENTIFIER OF THE CREATED DATA FILE
C
      DATA RL/"LBINP","LBOUT","LINPG","LINPB","LR","LX","LOUTG","LOUTB",
     1"LTAP"/,RB/"  BNR","BTYPE","BVMOD","BVARG","BGP","BLP","BLQ","BSTL
     2"/
      WRITE (OTPT,10) NB,(HDLN(I),I=1,8)
      WRITE (OTPT,70) NB,NTL
      WRITE (OTPT,20)
      WRITE (OTPT,30) (RL(I),I=1,9)
      WRITE (OTPT,50) (LBINP(I),LBOUT(I),LINPG(I),LINPB(I),LR(I),LX(I),L
     1OUTG(I),LOUTB(I),LTAP(I),I=1,NTL)
      WRITE (OTPT,40)
      WRITE (OTPT,30) (RB(I),I=1,8)
      WRITE (OTPT,60) (BNR(I),BTYP(I),BVMOD(I),BVARG(I),BGP(I),BLP(I),BL
     1Q(I),BSTL(I),I=1,NB)
      RETURN
   10 FORMAT (//1X,"B",I3.3,8A10/)
   20 FORMAT (//1X,"LINE DATA"/)
   30 FORMAT (2(1X,A5),7(6X,A5,4X))
   40 FORMAT (//1X,"BUS DATA"/)
   50 FORMAT ((2(1X,I5),2X,7(2X,E13.7)))
   60 FORMAT ((2(1X,I5),2X,6(2X,E13.7)))
   70 FORMAT (1X,"NB = ",I3.3,",     ","NTL =",I4.3)
      END
```

```
C
C
C
      SUBROUTINE FORMPR (LBINP,LBOUT,BTYP,YT,JRYT,ICYT,BCV,V,WS,LWS,NB,N
     1TL,NLB,IP,INPT,OTPT,IFLAG,IWRITE)
C
C     14.08.1982.
C
C     SUBROUTINE  FORMPR FORMULATES THE LOAD FLOW PROBLEM, I.E.,
C     THE SPARSE BUS ADMITTANCE  MATRIX  (VECTORS YT,JRYT,ICYT),
C     THE RIGHT HAND SIDE VECTOR BCV OF POWER FLOW EQUATIONS AND
C     VECTOR V OF THE INITIAL BUS VOLTAGES
C
      INTEGER LBINP(1),LBOUT(1),JRYT(1),ICYT(1),BTYP(1),OTPT
      REAL BCV(1),WS(1)
      COMPLEX YT(1),V(1)
      COMMON /MDFRMPR/ JINPG,JINPB,JLG,JLB,JOUTG,JOUTB,JTAP,JNR,JVM,JVA,
     1JGP,JLP,JLQ,JSTL,JMX
      IFLAG=0
      JINPG=1
      JINPB=JINPG+NTL
      JLG=JINPB+NTL
      JLB=JLG+NTL
      JOUTG=JLB+NTL
      JOUTB=JOUTG+NTL
      JTAP=JOUTB+NTL
      JNR=JTAP+NTL
      JVM=JNR+NB
      JVA=JVM+NB
      JGP=JVA+NB
      JLP=JGP+NB
      JLQ=JLP+NB
      JSTL=JLQ+NB
      JMX=JSTL+NB
      IF (JMX.GE.LWS) GO TO 50
      CALL RDAT (LBINP,LBOUT,WS(JINPG),WS(JINPB),WS(JLG),WS(JLB),WS(JOUT
     1G),WS(JOUTB),WS(JTAP),WS(JNR),BTYP,WS(JVM),WS(JVA),WS(JGP),WS(JLP)
     2,WS(JLQ),WS(JSTL),JRYT,NB,NTL,NLB,INPT,IWRITE)
      CALL FORMYT (LBINP,LBOUT,WS(JINPG),WS(JINPB),WS(JLG),WS(JLB),WS(JO
     1UTG),WS(JOUTB),WS(JTAP),WS(JSTL),JRYT,ICYT,YT,NB,NTL,NYT,OTPT,IWRI
     2TE)
      CALL FORMU (BTYP,WS(JVM),WS(JVA),WS(JGP),WS(JLP),WS(JLQ),BCV,NB,OT
     1PT,IWRITE)
      K1=JVM-1
      K2=JVA-1
      IF (IP.EQ.1) GO TO 20
      DO 10 I=1,NB
      R1=WS(K1+I)
      R2=WS(K2+I)
      V(I)=CMPLX(R1*COS(R2),R1*SIN(R2))
   10 CONTINUE
      GO TO 40
   20 DO 30 I=1,NB
      V(I)=CMPLX(WS(K1+I),WS(K2+I))
   30 CONTINUE
   40 RETURN
   50 IFLAG=-1
      RETURN
      END
```

```
C
C
C
      SUBROUTINE FORMTA (YT,JRYT,ICYT,BTYP,T,IRT,ICT,N,NT,OTPT,IWRITE)
C
C     07.05.1982.
C
C     SUBROUTINE FORMTA FORMS A REAL, APPROXIMATE ADJOINT MATRIX
C     OF THE POWER SYSTEM AND STORES IT IN A SPARSE FORM
C
      INTEGER JRYT(1),ICYT(1),BTYP(1),ICT(1),IRT(1),OTPT
      REAL T(1)
      COMPLEX YT(1),Y
      NT=0
      IF (IWRITE.GE.4) WRITE (OTPT,60)
      DO 50 I=1,N
C
C     MATRIX YT IS ANALYSED  ROW BY ROW  IN THIS LOOP.  K1 (OR K2)
C     INDICATES THE LOCATION OF THE FIRST (OR THE LAST) ELEMENT OF
C     THE ITH ROW OF NODAL  ADMITTANCE  MATRIX OF THE POWER SYSTEM
C
      K1=JRYT(I)
      K2=JRYT(I+1)-1
      JE=NT+1
      IR=2*I-1
      IR1=IR+1
C
C     IR, IR1  ARE INDICES OF ROWS OF THE ADJOINT MATRIX  CORRESPONDING
C     TO THE REAL AND IMAGINARY PARTS OF CURRENT OF THE ITH ADJOINT BUS
C
      ITYP=BTYP(I)
      IF (ITYP.EQ.1) GO TO 20
      DO 10 J=K1,K2
C
C     IN THIS LOOP TWO ROWS OF THE ADJOINT MATRIX CORRESPONDING TO AN
C     ADJOINT LOAD BUS ARE FORMED
C
      ITYP1=BTYP(ICYT(J))
      IF (ITYP1.EQ.2) GO TO 10
      IC=2*ICYT(J)-1
      IC1=IC+1
C
C     IC, IC1 ARE INDICES OF COLUMNS OF THE ADJOINT MATRIX ASSOCIATED
C     WITH THE REAL AND IMAGINARY  PARTS OF VOLTAGE OF AN ADJOINT BUS
C     ADJACENT TO THE ITH ADJOINT BUS
C
      R2=AIMAG(YT(J))
      R1=REAL(YT(J))
      CALL PTELT (T,NT,R1,IRT,ICT,IR,IC)
      CALL PTELT (T,NT,R2,IRT,ICT,IR1,IC)
      IF (ITYP1.EQ.1) GO TO 10
      CALL PTELT (T,NT,-R2,IRT,ICT,IR,IC1)
      CALL PTELT (T,NT,R1,IRT,ICT,IR1,IC1)
   10 CONTINUE
      GO TO 40
   20 DO 30 J=K1,K2
C
C     IN THIS LOOP A ROW OF THE ADJOINT MATRIX  CORRESPONDING TO THE
C     REAL PART OF CURRENT OF AN ADJOINT GENERATOR BUS IS DETERMINED
C
      ITYP1=BTYP(ICYT(J))
      IF (ITYP1.EQ.2) GO TO 30
      IC=2*ICYT(J)-1
      IC1=IC+1
C
```

```
C     IC, IC1 ARE INDICES OF COLUMNS OF THE ADJOINT MATRIX ASSOCIATED
C     WITH THE REAL AND IMAGINARY  PARTS OF VOLTAGE OF AN ADJOINT BUS
C
      Y=YT(J)
      CALL PTELT (T,NT,AIMAG(Y),IRT,ICT,IR,IC)
      IF (ITYP1.EQ.0) CALL PTELT (T,NT,REAL(Y),IRT,ICT,IR,IC1)
   30 CONTINUE
C
C     A ROW OF THE ADJOINT  MATRIX  CORRESPONDING  TO THE IMAGINARY
C     PART OF CURRENT OF AN ADJOINT GENERATOR BUS IS NOW DETERMINED
C
      CALL PTELT (T,NT,1.,IRT,ICT,IR1,IR1)
   40 IF (IWRITE.LT.4) GO TO 50
      WRITE (OTPT,70) I
      WRITE (OTPT,80) (IRT(J),ICT(J),T(J),J=JE,NT)
   50 CONTINUE
      RETURN
   60 FORMAT (//" MATRIX T OF ADJOINT EQUATIONS (APPROXIMATE VERSION)"/"
     1 VALUE OF AN ELEMENT IS PRECEDED BY THE ROW AND COLUMN INDICES")
   70 FORMAT (/" BUS NO.",I3)
   80 FORMAT (5(3X,I3,",",I3,":  ",E13.7))
      END
```

```
C
C
C
      SUBROUTINE FORMTAD (YT,JRYT,ICYT,BTYP,T,IRT,ICT,N,NT,OTPT,IWRITE)
C
C     10.05.1982.
C
C     SUBROUTINE FORMTAD FORMS A REAL, APPROXIMATE, DECOUPLED ADJOINT
C     MATRIX OF THE POWER SYSTEM AND STORES IT IN A SPARSE FORM
C
      INTEGER JRYT(1),ICYT(1),BTYP(1),ICT(1),IRT(1),OTPT
      REAL T(1)
      COMPLEX YT(1)
      NT=0
      IF (IWRITE.GE.4) WRITE (OTPT,60)
      DO 50 I=1,N
C
C     MATRIX YT IS ANALYSED  ROW BY ROW  IN THIS LOOP.  K1 (OR K2)
C     INDICATES THE LOCATION OF THE FIRST (OR THE LAST) ELEMENT OF
C     THE ITH ROW OF NODAL  ADMITTANCE  MATRIX OF THE POWER SYSTEM
C
      K1=JRYT(I)
      K2=JRYT(I+1)-1
      JE=NT+1
      IR=2*I-1
      IR1=IR+1
C
C     IR, IR1  ARE INDICES OF ROWS OF THE ADJOINT MATRIX  CORRESPONDING
C     TO THE REAL AND IMAGINARY PARTS OF CURRENT OF THE ITH ADJOINT BUS
C
      ITYP=BTYP(I)
      IF (ITYP.EQ.1) GO TO 20
      DO 10 J=K1,K2
C
C     IN THIS LOOP TWO ROWS OF THE ADJOINT MATRIX CORRESPONDING TO AN
C     ADJOINT LOAD BUS ARE FORMED
C
      ITYP1=BTYP(ICYT(J))
      IF (ITYP1.EQ.2) GO TO 10
      IC=2*ICYT(J)-1
      IC1=IC+1
C
C     IC, IC1 ARE INDICES OF COLUMNS OF THE ADJOINT MATRIX ASSOCIATED
C     WITH THE REAL AND IMAGINARY  PARTS OF VOLTAGE OF AN ADJOINT BUS
C     ADJACENT TO THE ITH ADJOINT BUS
C
      R2=AIMAG(YT(J))
      CALL PTELT (T,NT,R2,IRT,ICT,IR1,IC)
      CALL PTELT (T,NT,-R2,IRT,ICT,IR,IC1)
   10 CONTINUE
      GO TO 40
   20 DO 30 J=K1,K2
C
C     IN THIS LOOP A ROW OF THE ADJOINT MATRIX  CORRESPONDING TO THE
C     REAL PART OF CURRENT OF AN ADJOINT GENERATOR BUS IS DETERMINED
C
      IF (BTYP(ICYT(J)).EQ.2) GO TO 30
      IC=2*ICYT(J)-1
C
C     IC IS THE INDEX OF COLUMN OF THE ADJOINT MATRIX ASSOCIATED
C     WITH THE REAL PART OF VOLTAGE OF AN ADJOINT BUS
C
      CALL PTELT (T,NT,AIMAG(YT(J)),IRT,ICT,IR,IC)
   30 CONTINUE
C
```

```
C      A ROW OF THE ADJOINT  MATRIX  CORRESPONDING  TO THE IMAGINARY
C      PART OF CURRENT OF AN ADJOINT GENERATOR BUS IS NOW DETERMINED
C
       CALL PTELT (T,NT,1.,IRT,ICT,IR1,IR1)
    40 IF (IWRITE.LT.4) GO TO 50
       WRITE (OTPT,70) I
       WRITE (OTPT,80) (IRT(J),ICT(J),T(J),J=JE,NT)
    50 CONTINUE
       RETURN
    60 FORMAT (//" MATRIX T OF ADJOINT EQUATIONS (APPROXIMATE, DECOUPLED
      1VERSION)"/" VALUE OF AN ELEMENT IS PRECEDED BY THE ROW AND COLUMN
      2INDICES")
    70 FORMAT (/" BUS NO.",I3)
    80 FORMAT (5(3X,I3,",",I3,": ",E13.7))
       END
```

```
C
C
C
      SUBROUTINE FORMTD (YT,JRYT,ICYT,BCS,V,BTYP,T,IRT,ICT,N,NT,OTPT,IWR
     1ITE)
C
C     07.05.1982.
C
C     SUBROUTINE FORMTD FORMS A REAL, DECOUPLED ADJOINT MATRIX
C     OF THE POWER SYSTEM AND STORES IT IN A SPARSE FORM
C
      INTEGER JRYT(1),ICYT(1),BTYP(1),ICT(1),IRT(1),OTPT
      REAL T(1)
      COMPLEX YT(1),BCS(1),V(1),VA,Y
      NT=0
      IF (IWRITE.GE.4) WRITE (OTPT,70)
      DO 60 I=1,N
C
C     MATRIX YT IS ANALYSED  ROW BY ROW  IN THIS LOOP.  K1 (OR K2)
C     INDICATES THE LOCATION OF THE FIRST (OR THE LAST) ELEMENT OF
C     THE ITH ROW OF NODAL  ADMITTANCE  MATRIX OF THE POWER SYSTEM
C
      K1=JRYT(I)
      K2=JRYT(I+1)-1
      JE=NT+1
      IR=2*I-1
      IR1=IR+1
C
C     IR, IR1  ARE INDICES OF ROWS OF THE ADJOINT MATRIX  CORRESPONDING
C     TO THE REAL AND IMAGINARY PARTS OF CURRENT OF THE ITH ADJOINT BUS
C
      ITYP=BTYP(I)
      IF (ITYP.EQ.1) GO TO 20
      DO 10 J=K1,K2
C
C     IN THIS LOOP TWO ROWS OF THE ADJOINT MATRIX CORRESPONDING TO AN
C     ADJOINT LOAD BUS ARE FORMED
C
      ITYP1=BTYP(ICYT(J))
      IF (ITYP1.EQ.2) GO TO 10
      IC=2*ICYT(J)-1
      IC1=IC+1
C
C     IC, IC1 ARE INDICES OF COLUMNS OF THE ADJOINT MATRIX ASSOCIATED
C     WITH THE REAL AND IMAGINARY  PARTS OF VOLTAGE OF AN ADJOINT BUS
C     ADJACENT TO THE ITH ADJOINT BUS
C
      R2=AIMAG(YT(J))
      CALL PTELT (T,NT,R2,IRT,ICT,IR1,IC)
      CALL PTELT (T,NT,-R2,IRT,ICT,IR,IC1)
   10 CONTINUE
      PSI=AIMAG(BCS(I)/V(I)**2)
      GO TO 40
   20 VA=V(I)
      DO 30 J=K1,K2
C
C     IN THIS LOOP A ROW OF THE ADJOINT MATRIX  CORRESPONDING TO THE
C     REAL PART OF CURRENT OF AN ADJOINT GENERATOR BUS IS DETERMINED
C
      IF (BTYP(ICYT(J)).EQ.2) GO TO 30
      IC=2*ICYT(J)-1
C
C     IC IS THE INDEX OF COLUMN OF THE ADJOINT MATRIX ASSOCIATED
C     WITH THE REAL PART OF VOLTAGE OF AN ADJOINT BUS
C
```

```
      Y=VA*YT(J)
      CALL PTELT (T,NT,AIMAG(Y),IRT,ICT,IR,IC)
   30 CONTINUE
C
C     A ROW OF THE ADJOINT  MATRIX  CORRESPONDING  TO THE IMAGINARY
C     PART OF CURRENT OF AN ADJOINT GENERATOR BUS IS NOW DETERMINED
C
      CALL PTELT (T,NT,REAL(VA),IRT,ICT,IR1,IR1)
      PSI=AIMAG(BCS(I)/VA)
   40 T(JE)=T(JE)+PSI
      IF (ITYP.EQ.1) GO TO 50
      JE=JE+1
      T(JE)=T(JE)+PSI
   50 IF (IWRITE.LT.4) GO TO 60
      IF (ITYP.EQ.0) JE=JE-1
      WRITE (OTPT,80) I
      WRITE (OTPT,90) (IRT(J),ICT(J),T(J),J=JE,NT)
   60 CONTINUE
      RETURN
   70 FORMAT (//" MATRIX T OF ADJOINT EQUATIONS (DECOUPLED VERSION)"/" V
     1ALUE OF AN ELEMENT IS PRECEDED BY THE ROW AND COLUMN INDICES")
   80 FORMAT (/" BUS NO.",I3)
   90 FORMAT (5(3X,I3,",",I3,": ",E13.7))
      END
```

```
C
C
C
      SUBROUTINE FORMTE (YT,JRYT,ICYT,BCS,V,BTYP,T,IRT,ICT,N,NT,OTPT,IWR
     1ITE)
C
C     06.05.1982.
C
C     SUBROUTINE FORMTE FORMS A REAL, EXACT ADJOINT MATRIX
C     OF THE POWER  SYSTEM AND  STORES IT IN A SPARSE FORM
C
      INTEGER JRYT(1),ICYT(1),BTYP(1),ICT(1),IRT(1),OTPT
      REAL T(1)
      COMPLEX YT(1),BCS(1),V(1),PSI,VA,Y
      NT=0
      IF (IWRITE.GE.4) WRITE (OTPT,70)
      DO 60 I=1,N
C
C     MATRIX YT IS ANALYSED  ROW BY ROW  IN THIS LOOP.  K1 (OR K2)
C     INDICATES THE LOCATION OF THE FIRST (OR THE LAST) ELEMENT OF
C     THE ITH ROW OF NODAL  ADMITTANCE  MATRIX OF THE POWER SYSTEM
C
      K1=JRYT(I)
      K2=JRYT(I+1)-1
      JE=NT+1
      IR=2*I-1
      IR1=IR+1
C
C     IR, IR1  ARE INDICES OF ROWS OF THE ADJOINT MATRIX  CORRESPONDING
C     TO THE REAL AND IMAGINARY PARTS OF CURRENT OF THE ITH ADJOINT BUS
C
      ITYP=BTYP(I)
      IF (ITYP.EQ.1) GO TO 20
      DO 10 J=K1,K2
C
C     IN THIS LOOP TWO ROWS OF THE ADJOINT MATRIX CORRESPONDING TO AN
C     ADJOINT LOAD BUS ARE FORMED
C
      IF (BTYP(ICYT(J)).EQ.2) GO TO 10
      IC=2*ICYT(J)-1
      IC1=IC+1
C
C     IC, IC1 ARE INDICES OF COLUMNS OF THE ADJOINT MATRIX ASSOCIATED
C     WITH THE REAL AND IMAGINARY  PARTS OF VOLTAGE OF AN ADJOINT BUS
C     ADJACENT TO THE ITH ADJOINT BUS
C
      R2=AIMAG(YT(J))
      R1=REAL(YT(J))
      CALL PTELT (T,NT,R1,IRT,ICT,IR,IC)
      CALL PTELT (T,NT,R2,IRT,ICT,IR1,IC)
      CALL PTELT (T,NT,-R2,IRT,ICT,IR,IC1)
      CALL PTELT (T,NT,R1,IRT,ICT,IR1,IC1)
   10 CONTINUE
      PSI=BCS(I)/V(I)**2
      GO TO 40
   20 VA=V(I)
      DO 30 J=K1,K2
C
C     IN THIS LOOP A ROW OF THE ADJOINT MATRIX  CORRESPONDING TO THE
C     REAL PART OF CURRENT OF AN ADJOINT GENERATOR BUS IS DETERMINED
C
      IF (BTYP(ICYT(J)).EQ.2) GO TO 30
      IC=2*ICYT(J)-1
      IC1=IC+1
C
```

```
C     IC, IC1 ARE INDICES OF COLUMNS OF THE ADJOINT MATRIX ASSOCIATED
C     WITH THE REAL AND IMAGINARY PARTS OF VOLTAGE OF AN ADJOINT BUS
C
      Y=VA*YT(J)
      CALL PTELT (T,NT,AIMAG(Y),IRT,ICT,IR,IC)
      CALL PTELT (T,NT,REAL(Y),IRT,ICT,IR,IC1)
   30 CONTINUE
C
C     A ROW OF THE ADJOINT  MATRIX  CORRESPONDING  TO THE IMAGINARY
C     PART OF CURRENT OF AN ADJOINT GENERATOR BUS  IS NOW DETERMINED
C
      CALL PTELT (T,NT,AIMAG(VA),IRT,ICT,IR1,IR)
      CALL PTELT (T,NT,REAL(VA),IRT,ICT,IR1,IR1)
      PSI=(BCS(I)/VA)*(0.,-1.)
   40 T(JE)=T(JE)+REAL(PSI)
      JE=JE+1
      T(JE)=T(JE)+AIMAG(PSI)
      IF (ITYP.EQ.1) GO TO 50
      JE=JE+1
      T(JE)=T(JE)+AIMAG(PSI)
      JE=JE+1
      T(JE)=T(JE)-REAL(PSI)
   50 IF (IWRITE.LT.4) GO TO 60
      JE=JE-1
      IF (ITYP.EQ.0) JE=JE-2
      WRITE (OTPT,80) I
      WRITE (OTPT,90) (IRT(J),ICT(J),T(J),J=JE,NT)
   60 CONTINUE
      RETURN
   70 FORMAT (//" MATRIX T OF ADJOINT EQUATIONS (EXACT VERSION)"/" VALUE
     1 OF AN ELEMENT IS PRECEDED BY THE ROW AND COLUMN INDICES")
   80 FORMAT (/" BUS NO.",I3)
   90 FORMAT (5(3X,I3,",",I3,":  ",E13.7))
      END
```

```
C
C
C
      SUBROUTINE FORMTM (YT,JRYT,ICYT,BCS,V,BTYP,T,IRT,ICT,N,NT,OTPT,IWR
     1ITE)
C
C     10.05.1982.
C
C     SUBROUTINE FORMTM FORMS A REAL, MIXED ADJOINT MATRIX
C     OF THE POWER  SYSTEM AND STORES IT IN A SPARSE  FORM
C
      INTEGER JRYT(1),ICYT(1),BTYP(1),ICT(1),IRT(1),OTPT
      REAL T(1)
      COMPLEX YT(1),BCS(1),V(1),PSI,VA,Y
      NT=0
      IF (IWRITE.GE.4) WRITE (OTPT,70)
      DO 60 I=1,N
C
C     MATRIX YT IS ANALYSED  ROW BY ROW  IN THIS LOOP.  K1 (OR K2)
C     INDICATES THE LOCATION OF THE FIRST (OR THE LAST) ELEMENT OF
C     THE ITH ROW OF NODAL  ADMITTANCE  MATRIX OF THE POWER SYSTEM
C
      K1=JRYT(I)
      K2=JRYT(I+1)-1
      JE=NT+1
      IR=2*I-1
      IR1=IR+1
C
C     IR, IR1  ARE INDICES OF ROWS OF THE ADJOINT MATRIX  CORRESPONDING
C     TO THE REAL AND IMAGINARY PARTS OF CURRENT OF THE ITH ADJOINT BUS
C
      ITYP=BTYP(I)
      IF (ITYP.EQ.1) GO TO 20
      DO 10 J=K1,K2
C
C     IN THIS LOOP TWO ROWS OF THE ADJOINT MATRIX CORRESPONDING TO AN
C     ADJOINT LOAD BUS ARE FORMED
C
      ITYP1=BTYP(ICYT(J))
      IF (ITYP1.EQ.2) GO TO 10
      IC=2*ICYT(J)-1
      IC1=IC+1
C
C     IC, IC1 ARE INDICES OF COLUMNS OF THE ADJOINT MATRIX ASSOCIATED
C     WITH THE REAL AND IMAGINARY  PARTS OF VOLTAGE OF AN ADJOINT BUS
C     ADJACENT TO THE ITH ADJOINT BUS
C
      R2=AIMAG(YT(J))
      R1=REAL(YT(J))
      CALL PTELT (T,NT,R1,IRT,ICT,IR,IC)
      CALL PTELT (T,NT,R2,IRT,ICT,IR1,IC)
      IF (ITYP1.EQ.1) GO TO 10
      CALL PTELT (T,NT,-R2,IRT,ICT,IR,IC1)
      CALL PTELT (T,NT,R1,IRT,ICT,IR1,IC1)
   10 CONTINUE
      PSI=BCS(I)/V(I)**2
      GO TO 40
   20 VA=V(I)
      DO 30 J=K1,K2
C
C     IN THIS LOOP A ROW OF THE ADJOINT MATRIX  CORRESPONDING TO THE
C     REAL PART OF CURRENT OF AN ADJOINT GENERATOR BUS IS DETERMINED
C
      ITYP1=BTYP(ICYT(J))
      IF (ITYP1.EQ.2) GO TO 30
```

```
      IC=2*ICYT(J)-1
      IC1=IC+1
C
C     IC, IC1 ARE INDICES OF COLUMNS OF THE ADJOINT MATRIX ASSOCIATED
C     WITH THE REAL AND IMAGINARY  PARTS OF VOLTAGE OF AN ADJOINT BUS
C
      Y=VA*YT(J)
      CALL PTELT (T,NT,AIMAG(Y),IRT,ICT,IR,IC)
      IF (ITYP1.EQ.0) CALL PTELT (T,NT,REAL(Y),IRT,ICT,IR,IC1)
   30 CONTINUE
C
C     A ROW OF THE ADJOINT  MATRIX  CORRESPONDING  TO THE IMAGINARY
C     PART OF CURRENT OF AN ADJOINT GENERATOR BUS IS NOW DETERMINED
C
      CALL PTELT (T,NT,REAL(VA),IRT,ICT,IR1,IR1)
      PSI=(BCS(I)/VA)*(0.,-1.)
   40 T(JE)=T(JE)+REAL(PSI)
      IF (ITYP.EQ.1) GO TO 50
      JE=JE+1
      T(JE)=T(JE)+AIMAG(PSI)
      JE=JE+1
      T(JE)=T(JE)+AIMAG(PSI)
      JE=JE+1
      T(JE)=T(JE)-REAL(PSI)
   50 IF (IWRITE.LT.4) GO TO 60
      IF (ITYP.EQ.0) JE=JE-3
      WRITE (OTPT,80) I
      WRITE (OTPT,90) (IRT(J),ICT(J),T(J),J=JE,NT)
   60 CONTINUE
      RETURN
   70 FORMAT (//" MATRIX T OF ADJOINT EQUATIONS (MIXED VERSION)"/" VALUE
     1 OF AN ELEMENT IS PRECEDED BY THE ROW AND COLUMN INDICES")
   80 FORMAT (/" BUS NO.",I3)
   90 FORMAT (5(3X,I3,",",I3,": ",E13.7))
      END
```

```
C
C
C
      SUBROUTINE FORMU (BTYP,BVMOD,BVARG,BGP,BLP,BLQ,BCV,NB,OTPT,IWRITE)
C
C     05.05.1982.
C
C     SUBROUTINE FORMU FORMS VECTOR BCV OF BUS CONTROL VARIABLES
C     OF THE POWER SYSTEM
C
      INTEGER BTYP(1),OTPT
      REAL BVMOD(1),BVARG(1),BGP(1),BLP(1),BLQ(1),BCV(1)
      DO 40 I=1,NB
      J=2*I-1
      IF (BTYP(I)-1) 10,20,30
C
C     SETTING UP VALUES OF ELEMENTS OF BCV  CORRESPONDING TO A LOAD BUS
C
   10 BCV(J)=BGP(I)-BLP(I)
      J=J+1
      BCV(J)=-BLQ(I)
      GO TO 40
C
C     SETTING UP VALUES OF ELEMENTS OF BCV CORRESPONDING TO A GENERATOR
C     BUS
C
   20 BCV(J)=BGP(I)-BLP(I)
      J=J+1
      BCV(J)=BVMOD(I)
      GO TO 40
C
C     SETTING UP VALUES OF ELEMENTS OF  BCV  CORRESPONDING TO SLACK BUS
C
   30 BCV(J)=BVMOD(I)
      J=J+1
      BCV(J)=BVARG(I)
   40 CONTINUE
      IF (IWRITE.LT.3) GO TO 50
      WRITE (OTPT,60)
      WRITE (OTPT,70) (I,BCV(2*I-1),BCV(2*I),I=1,NB)
   50 RETURN
   60 FORMAT (//" VECTOR BCV OF BUS CONTROL VARIABLES"/" VALUE OF AN ELE
     1MENT IS PRECEDED BY THE NUMBER OF BUS"/)
   70 FORMAT (3(3X,I3,":",2(1X,E13.7)))
      END
```

```
C
C
C
      SUBROUTINE FORMYT (LBINP,LBOUT,LINPG,LINPB,LG,LB,LOUTG,LOUTB,LTAP,
     1BSTL,JRYT,ICYT,YT,NB,NTL,NYT,OTPT,IWRITE)
C
C     05.05.1982.
C
C     SUBROUTINE FORMYT FORMS THE NODAL ADMITTANCE MATRIX OF POWER
C     SYSTEM AND STORES IT IN A SPARSE FORM (VECTORS YT,JRYT,ICYT)
C
      INTEGER LBINP(1),LBOUT(1),JRYT(1),ICYT(1),OTPT
      REAL LINPG(1),LINPB(1),LG(1),LB(1),LOUTG(1),LOUTB(1),LTAP(1),BSTL(
     11)
      COMPLEX YT(1),Y
      NYT=NB+2*NTL
C
C     BUS STATIC LOADS ARE PLACED INTO THE NODAL ADMITTANCE MATRIX
C
      DO 10 I=1,NB
      J=JRYT(I)
      YT(J)=CMPLX(0.,BSTL(I))
      ICYT(NYT+I)=J
   10 CONTINUE
C
C     LINE PARAMETERS ARE ADDED TO THE NODAL ADMITTANCE MATRIX
C
      DO 20 I=1,NTL
      IB1=LBINP(I)
      IB2=LBOUT(I)
      Y=CMPLX(LG(I),LB(I))
      L1=JRYT(IB1)
      L2=JRYT(IB2)
      YT(L1)=YT(L1)+CMPLX(LINPG(I),LINPB(I))+Y/(LTAP(I)**2)
      YT(L2)=YT(L2)+CMPLX(LOUTG(I),LOUTB(I))+Y
      ICYT(L1)=IB1
      ICYT(L2)=IB2
      K1=NYT+IB1
      K2=NYT+IB2
      ICYT(K1)=ICYT(K1)+1
      ICYT(K2)=ICYT(K2)+1
      L1=ICYT(K1)
      L2=ICYT(K2)
      Y=-Y/LTAP(I)
      YT(L1)=Y
      YT(L2)=Y
      ICYT(L1)=IB2
      ICYT(L2)=IB1
   20 CONTINUE
      IF (IWRITE.LT.3) GO TO 40
      WRITE (OTPT,50)
      DO 30 I=1,NB
      K1=JRYT(I)
      K2=JRYT(I+1)-1
      WRITE (OTPT,60) ICYT(K1)
      WRITE (OTPT,70) (ICYT(J),YT(J),J=K1,K2)
   30 CONTINUE
   40 RETURN
   50 FORMAT (//" BUS ADMITTANCE MATRIX YT"/" IN EACH ROW DATA IS IN SEQ
     1UENCE: COLUMN INDEX,REAL(YT),IMAG(YT)")
   60 FORMAT (/" BUS NO.",I3)
   70 FORMAT (3(3X,I3,":",2(1X,E13.7)))
      END
```

```
C
C
C
      SUBROUTINE LFLFD1M (NB,NLB,NYT,JRYT,ICYT,BTYP,YT,V,BCV,W,LW,ITEL,V
     1EPS,TIMEL,MODE,IFLAG,OTPT,IWRITE)
C
C     04.08.1982.
C
C     SUBROUTINE LFLFD1M IS THE HIGHEST LEVEL SUBROUTINE FOR SOLVING
C     THE LOAD FLOW PROBLEM USING A SPARSE MATRIX TECHNIQUE (HARWELL
C     PACKAGE MA28) AND THE FAST DECOUPLED METHOD
C
C     LIBRARY : LIBRHSM (HARWELL PACKAGE MA28)
C
C     NB      NUMBER OF BUSES
C     NLB     NUMBER OF LOAD BUSES
C     NYT     NUMBER OF NONZEROS IN THE BUS ADMITTANCE MATRIX
C     JRYT    VECTOR OF LENGTH (NB+1). IT HOLDS ROW INDICES OF THE
C             SPARSE BUS ADMITTANCE MATRIX
C     ICYT    VECTOR OF LENGTH NYT. IT HOLDS COLUMN INDICES OF THE
C             SPARSE BUS ADMITTANCE MATRIX
C     BTYP    VECTOR OF BUS TYPES  (0  LOAD BUS, 1  GENERATOR BUS)
C     YT      SPARSE BUS ADMITTANCE MATRIX
C     V       COMPLEX BUS VOLTAGES (RECTANGULAR COORDINATES)
C     BCV     REAL VECTOR OF THE LENGTH 2*(NB-1). IT HOLDS NOMINAL
C             VALUES OF BUS CONTROL VARIABLES
C     W       REAL WORKSPACE
C     LW      LENGTH OF THE WORSPACE W
C     ITEL    LIMIT OF ITERATIONS
C     VEPS    REQUIRED ACCURACY OF SOLUTION
C     TIMEL   LIMIT OF ITERATION TIME
C     MODE    MODE OF OPERATION :
C              0 - EVALUATE AND FACTORIZE  APPROXIMATE JACOBIAN MATRICES
C              1 - PERFORM  P-DELTA  ITERATION (FOR FACTORIZED MATRICES)
C              2 - PERFORM  Q-MOD.V  ITERATION (FOR FACTORIZED MATRICES)
C              3 - PERFORM BOTH ITERATIONS (FOR FACTORIZED MATRICES)
C     IFLAG   RETURN FLAG :
C             -2 - INCORRECT USAGE  (SINGULAR P-DELTA OR Q-V MATRIX)
C             -1 - INCORRECT PARAMETERS
C              0 - NORMAL RETURN (REQUIRED ACCURACY OBTAINED)
C              1 - LIMIT OF ITERATIONS REACHED
C              2 - LIMIT OF ITERATION TIME REACHED
C     OTPT    NUMBER OF OUTPUT UNIT
C
      INTEGER JRYT(1),ICYT(1),BTYP(1)
      REAL W(1),BCV(1)
      COMPLEX YT(1),V(1)
      COMMON /MDLFLF/ JA1,JICN1,JIKP1,JA2,JICN2,JIKP2,JVM,JVA,NZ,NZR
      IF (MODE.LT.0.OR.MODE.GT.3) GO TO 20
      IF (MODE.NE.0) GO TO 10
      N=NB-1
      NZ1=NYT
      NZ2=(NYT*(NLB**2))/(N**2)
      LICN1=3*NZ1
      LICN2=3*NZ2
      LIRN1=NZ1+N+N
      LIRN2=NZ2+NLB+NLB
      JNRB=1
      JA1=JNRB+NB
      JICN1=JA1+LICN1
      JIKP1=JICN1+LICN1
      JA2=JIKP1+5*N
      JICN2=JA2+LICN2
      JIKP2=JICN2+LICN2
      JRHS=JIKP2+5*NLB
```

```
      JW=JRHS+NB
      JVM=JW+NB
      JVA=JVM+NB
      MXJ=JVA+NB
      JIRN2=JRHS
      JIW2=JIRN2+LIRN2
      JW2=JIW2+8*NLB
      JIRN1=JA2
      JIW1=JIRN1+LIRN1
      JW1=JIW1+8*N
      MXJ=MAX0(MXJ,JW1+N,JW2+NLB)
      J=LW-MXJ
      IF (J.LT.0) GO TO 20
   10 CALL LFLFDAM (NB,NLB,NZ,NZR,JRYT,ICYT,BTYP,YT,V,BCV,LICN1,LIRN1,LI
     1CN2,LIRN2,W(JA1),W(JICN1),W(JIKP1),W(JA2),W(JICN2),W(JIKP2),W(JRHS
     2),W(JW),W(JVM),W(JVA),W(JNRB),W(JIRN1),W(JIW1),W(JW1),W(JIRN2),W(J
     3IW2),W(JW2),MODE,ITEL,VEPS,TIMEL,IFLAG,OTPT,IWRITE)
C
C      IN THE ABOVE CALL FORMAL PARAMETER ARRAYS: (RHS,W,VM,VA)
C                 AND    (IRN2,IW2,W2)
C            AS WELL AS (A2,ICN2,IKEEP2,IRN2,IW2,W2)
C                 AND    (IRN1,IW1,W1)
C            SHARE THE SAME WORKSPACE
C
      RETURN
   20 IFLAG=-1
      RETURN
      END
```

```
C
C
C
      SUBROUTINE LFLFDAM (NB,NLB,NZ,NZR,JRYT,ICYT,BTYP,YT,V,BCV,LICN1,LI
     1RN1,LICN2,LIRN2,A1,ICN1,IKEEP1,A2,ICN2,IKEEP2,RHS,W,VM,VA,NRB,IRN1
     2,IW1,W1,IRN2,IW2,W2,MODE,ITE,VEPS,TIMEX,IERR,OTPT,IWRITE)
C
C     05.08.1982.
C
C     THIS SUBROUTINE IMPLEMENTS THE FAST DECOUPLED ITERATIVE METHOD
C     USING THE HARWELL PACKAGE MA28 FOR  SOLVING THE SPARSE SYSTEMS
C     OF LINEAR EQUATIONS WITH REAL COEFFICIENTS
C
C     LIBRARY : LIBRHSM (HARWELL PACKAGE MA28)
C
      INTEGER JRYT(1),ICYT(1),BTYP(1),ICN1(1),ICN2(1),IKEEP1(1),IKEEP2(1
     1),IW1(1),IW2(1),NRB(1),IRN1(1),IRN2(1),OTPT
      REAL BCV(1),A1(1),A2(1),RHS(1),W(1),W1(1),W2(1),VM(1),VA(1),R(2)
      COMPLEX YT(1),V(1),VV
      LOGICAL SWITCH
      DATA R/"P-DELTA","    Q-V"/
      IF (IWRITE.LT.1) GO TO 10
      WRITE (OTPT,230) NB
   10 CALL SECOND (TTIME1)
      T1=TTIME1
      N=NB-1
      IERR=-2
      IF (MODE.NE.0) GO TO 100
C
C     SET ORDERING OF BUSES
C
      L=0
      K=NLB
      DO 30 I=1,N
      IF (BTYP(I).NE.0) GO TO 20
      L=L+1
      NRB(I)=L
      GO TO 30
   20 K=K+1
      NRB(I)=K
   30 CONTINUE
C
C     SET AND FACTORIZE SPARSE MATRICES OF REAL COEFFICIENTS
C
      NZ=0
      IF (IWRITE.LT.4) GO TO 40
      WRITE (OTPT,240)
   40 DO 60 I=1,N
      NZ=NZ+1
      L=NZ
      J1=JRYT(I)+1
      J2=JRYT(I+1)-1
      X=0.0
      DO 50 J=J1,J2
      KK=ICYT(J)
      IF (YT(J).EQ.(0.,0.)) GO TO 50
      XX=1.0/AIMAG(1.0/YT(J))
      X=X+XX
      IF (BTYP(KK).EQ.2) GO TO 50
      NZ=NZ+1
      ICN1(NZ)=KK
      IRN1(NZ)=I
      A1(NZ)=XX
   50 CONTINUE
      ICN1(L)=I
```

```
      IRN1(L)=I
      A1(L)=-X
      IF (IWRITE.LT.4) GO TO 60
      WRITE (OTPT,250) I
      WRITE (OTPT,260) (IRN1(J),ICN1(J),A1(J),J=L,NZ)
   60 CONTINUE
      U=0.1
      CALL MA28A (N,NZ,A1,LICN1,IRN1,LIRN1,ICN1,U,IKEEP1,IW1,W1,IFLAG)
      IF (IFLAG.LT.0) RETURN
      NZR=0
      IF (IWRITE.LT.4) GO TO 70
      WRITE (OTPT,270)
   70 DO 90 I=1,N
      K=BTYP(I)
      IF (K.NE.0) GO TO 90
      L=NZR+1
      LL=NRB(I)
      J1=JRYT(I)
      J2=JRYT(I+1)-1
      DO 80 J=J1,J2
      KK=ICYT(J)
      IF (BTYP(KK).NE.0) GO TO 80
      NZR=NZR+1
      ICN2(NZR)=NRB(KK)
      IRN2(NZR)=LL
      A2(NZR)=-AIMAG(YT(J))
   80 CONTINUE
      IF (IWRITE.LT.4) GO TO 90
      WRITE (OTPT,250) I
      WRITE (OTPT,260) (IRN2(J),ICN2(J),A2(J),J=L,NZR)
   90 CONTINUE
      CALL MA28A (NLB,NZR,A2,LICN2,IRN2,LIRN2,ICN2,U,IKEEP2,IW2,W2,IFLAG
     1)
      IF (IFLAG.LT.0) RETURN
C
C     SET INITIAL VALUES AND CHECK LIMIT OF ITERATIONS
C
  100 IT=0
      CMX=0.0
      CALL SECOND (TTIME2)
      IF (ITE.EQ.0) GO TO 180
C
C     CONVERT VOLTAGES TO POLAR COORDINATES
C
      DO 110 I=1,NB
      VV=V(I)
      VM(I)=CABS(VV)
      VA(I)=ATAN2(AIMAG(VV),REAL(VV))
  110 CONTINUE
C
C     ITERATION LOOP
C
      CORRA=0.0
      CORRM=0.0
      IF (MODE.EQ.1) GO TO 120
      IF (MODE.EQ.2) GO TO 140
  120 IT=IT+1
      CALL LFLFDBM (N,YT,V,VM,VA,BCV,JRYT,ICYT,A1,LICN1,ICN1,IKEEP1,RHS,
     1W,CORRA)
      IF (IWRITE.LT.2) GO TO 130
      CALL SECOND (T2)
      T=T2-T1
      WRITE (OTPT,280) IT,R(1),CORRA,T
      T1=T2
      IF (IWRITE.LT.3) GO TO 130
```

```
      WRITE (OTPT,290)
      WRITE (OTPT,300) (L,V(L),L=1,N)
  130 SWITCH=.TRUE.
      IF (IT.EQ.1.AND.(MODE.EQ.0.OR.MODE.EQ.3)) GO TO 140
      GO TO 150
  140 IT=IT+1
      CALL LFLFDCM (N,NLB,YT,V,VM,VA,BCV,JRYT,ICYT,BTYP,NRB,A2,LICN2,ICN
     12,IKEEP2,RHS,W,CORRM)
      SWITCH=.FALSE.
      IF (IWRITE.LT.2) GO TO 150
      CALL SECOND (T2)
      T=T2-T1
      WRITE (OTPT,280) IT,R(2),CORRM,T
      T1=T2
      IF (IWRITE.LT.3) GO TO 150
      WRITE (OTPT,300) (L,V(L),L=1,N)
      T1=T2
  150 CMX=AMAX1(CORRA,CORRM)
      CALL SECOND (TTIME2)
      IF (CMX.LE.VEPS) GO TO 200
      IF (TIMEX.LE.0.0) GO TO 160
      IF (TTIME2-TTIME1.GE.TIMEX) GO TO 190
  160 IF (ITE.LT.0) GO TO 170
      IF (IT.GE.ITE) GO TO 180
  170 IF (CORRA.GT.2.0*CORRM) GO TO 120
      IF (CORRM.GT.2.0*CORRA) GO TO 140
      IF (SWITCH) GO TO 140
      GO TO 120
  180 IERR=1
      GO TO 210
  190 IERR=2
      GO TO 210
  200 IERR=0
  210 ITE=IT
      TIMEX=TTIME2-TTIME1
      VEPS=CMX
      IF (IWRITE.LT.1) GO TO 220
      WRITE (OTPT,310) ITE,IERR,VEPS,TIMEX
      IF (IWRITE.LT.2) GO TO 220
      WRITE (OTPT,320)
      WRITE (OTPT,330) (L,V(L),VM(L),VA(L),L=1,NB)
  220 RETURN
  230 FORMAT (///" LOAD FLOW SOLUTION OF ",I3,"-BUS SYSTEM USING THE FAS
     1T DECOUPLED METHOD")
  240 FORMAT (//" MATRIX OF P-DELTA EQUATIONS"/" VALUE OF AN ELEMENT IS
     1PRECEDED BY THE ROW AND COLUMN INDICES")
  250 FORMAT (/" BUS NO. ",I3)
  260 FORMAT (5(3X,I3,",",I3,": ",E13.7))
  270 FORMAT (//" MATRIX OF Q-V EQUATIONS"/" VALUE OF AN ELEMENT IS PREC
     1EDED BY THE ROW AND COLUMN INDICES")
  280 FORMAT (//" RESULTS OF ITERATION NO.",3X,I3/" ITERATION TYPE:",8X,
     1A7/" ACCURACY OBTAINED:",2X,E10.4/" ITERATION TIME:",9X,F6.3," SEC
     2ONDS")
  290 FORMAT (/" VECTOR OF BUS VOLTAGES")
  300 FORMAT (3(3X,I4,":",2(1X,E13.7)))
  310 FORMAT (///" RESULTS OF ANALYSIS"//" NUMBER OF ITERATIONS:",6X,I3/
     1" RETURN FLAG:",16X,I2/" ACCURACY OBTAINED:",2X,E10.4/" ANALYSIS T
     2IME:",10X,F6.3," SECONDS")
  320 FORMAT (/" VECTOR OF BUS VOLTAGES"/4X,"BUS",7X,"RECTANGULAR COORDI
     1NATES",10X," POLAR COORDINATES"/)
  330 FORMAT ((3X,I4,2(3X,2(1X,E13.7))))
      END
```

```
C
C
C
      SUBROUTINE LFLFDBM (N,YT,V,VM,VA,BCV,JRYT,ICYT,A1,LICN1,ICN1,IKEEP
     11,RHS,W,CORRA)
C
C     05.08.1982.
C
C     SUBROUTINE LFLFDBM  DETERMINES THE RIGHT HAND SIDE VECTOR FOR
C     ARGUMENT CORRECTIONS OF THE FAST DECOUPLED METHOD, SOLVES THE
C     SPARSE SYSTEM OF LINEAR  EQUATIONS  AND UPDATES  THE VOLTAGES
C
C     LIBRARY : LIBRHSM  (HARWELL PACKAGE MA28)
C
      INTEGER JRYT(1),ICYT(1),ICN1(1),IKEEP1(1)
      REAL VM(1),VA(1),A1(1),BCV(1),RHS(1),W(1)
      COMPLEX YT(1),V(1),CURR,PW
C
C     SETTING RIGHT HAND SIDE VECTOR
C
      DO 10 I=1,N
      PW=V(I)*CONJG(CURR(V,YT,JRYT,ICYT,I))
      RHS(I)=(BCV(2*I-1)-REAL(PW))/VM(I)
   10 CONTINUE
C
C     SOLVING LINEAR EQUATIONS
C
      CALL MA28C (N,A1,LICN1,ICN1,IKEEP1,RHS,W,1)
C
C     UPDATING VOLTAGES
C
      CORRA=0.0
      DO 20 I=1,N
      V1=VM(I)
      DV=ABS(V1*(RHS(I)))
      V2=VA(I)+RHS(I)
      VA(I)=V2
      IF (DV.GT.CORRA) CORRA=DV
      V(I)=CMPLX(V1*COS(V2),V1*SIN(V2))
   20 CONTINUE
      RETURN
      END
```

```
C
C
C
      SUBROUTINE LFLFDCM (N,NLB,YT,V,VM,VA,BCV,JRYT,ICYT,BTYP,NRB,A2,LIC
     1N2,ICN2,IKEEP2,RHS,W,CORRM)
C
C     05.08.1982.
C
C     SUBROUTINE LFLFDCM DETERMINES THE RIGHT HAND SIDE VECTOR FOR
C     MODULUS CORRECTIONS OF THE FAST DECOUPLED METHOD, SOLVES THE
C     SPARSE  SYSTEM OF LINEAR  EQUATIONS AND UPDATES THE VOLTAGES
C
C     LIBRARY : LIBRHSM  (HARWELL PACKAGE MA28)
C
      INTEGER JRYT(1),ICYT(1),BTYP(1),NRB(1),ICN2(1),IKEEP2(1)
      REAL VM(1),VA(1),A2(1),BCV(1),RHS(1),W(1)
      COMPLEX YT(1),V(1),CURR,PW
C
C     SETTING RIGHT HAND SIDE VECTOR
C
      DO 10 I=1,N
      IF (BTYP(I).NE.0) GO TO 10
      L=NRB(I)
      PW=V(I)*CONJG(CURR(V,YT,JRYT,ICYT,I))
      RHS(L)=(BCV(2*I)-AIMAG(PW))/VM(I)
   10 CONTINUE
C
C     SOLVING LINEAR EQUATIONS
C
      CALL MA28C (NLB,A2,LICN2,ICN2,IKEEP2,RHS,W,1)
C
C     UPDATE VOLTAGES
C
      CORRM=0.0
      DO 20 I=1,N
      IF (BTYP(I).NE.0) GO TO 20
      K=NRB(I)
      DV=ABS(RHS(K))
      IF (DV.GT.CORRM) CORRM=DV
      V1=VM(I)
      V2=V1+RHS(K)
      VM(I)=V2
      V(I)=(V2/V1)*V(I)
   20 CONTINUE
      RETURN
      END
```

```
C
C
C
      SUBROUTINE LFTTM (NB,NTL,JRYT,ICYT,BTYP,YT,V,BCV,W,LW,IVT,IP,ITEL,
     1VEPS,TIMEL,MODE,IFLAG,OTPT,IWRITE)
C
C     28.08.1982.
C
C     SUBROUTINE LFTTM IS THE HIGHEST LEVEL SUBROUTINE FOR SOLVING
C     THE  LOAD  FLOW  PROBLEM  USING  A SPARSE  MATRIX  TECHNIQUE
C     (HARWELL  PACKAGE  MA28)  AND  THE TELLEGEN  THEOREM  METHOD
C
C     LIBRARY : LIBRHSM (HARWELL PACKAGE MA28)
C
C     NB      NUMBER OF BUSES
C     NTL     NUMBER OF TRANSMISSION LINES
C     JRYT    VECTOR OF LENGTH (NB+1).  IT HOLDS ROW INDICES OF THE
C             SPARSE BUS ADMITTANCE MATRIX
C     ICYT    VECTOR OF LENGTH NB+2*NTL. IT HOLDS COLUMN INDICES OF
C             THE SPARSE BUS ADMITTANCE MATRIX
C     BTYP    VECTOR OF BUS TYPES  (0 LOAD BUS, 1 GENERATOR BUS)
C     YT      SPARSE BUS ADMITTANCE MATRIX
C     V       COMPLEX BUS VOLTAGES
C     BCV     REAL VECTOR OF THE LENGTH 2*(NB-1).  IT HOLDS NOMINAL
C             VALUES OF BUS CONTROL VARIABLES
C     W       REAL WORKSPACE
C     LW      LENTGH OF THE WORKSPACE W.  IT SHOULD  BE DECLARED AT
C             LEAST 70*NB+56*NTL-35
C     IVT     VERSION OF ADJOINT MATRIX WHICH IS TO BE USED.  RANGE
C             OF VALUES OF IVT IS FROM 1 TO 5
C     IP      VERSION OF THE POWER FLOW EQUATIONS (1 POLAR VERSION,
C             0 RECTANGULAR VERSION)
C     ITEL    LIMIT OF ITERATIONS
C     VEPS    REQUIRED ACCURACY OF SOLUTION
C     TIMEL   LIMIT OF ITERATION TIME
C     MODE    MODE OF OPERATION :
C             1-ADJOINT  MATRIX IS FORMED AND DECOMPOSED INTO LU FACTORS
C               IN THE FIRST  ITERATION OF THE CURRENT  CALL TO THE SUB-
C               ROUTINE AND IS UPDATED  AND FACTORIZED IN THE SUBSEQUENT
C               ITERATIONS USING THE PIVOTAL STRATEGY DETERMINED EARLIER
C             2-ADJOINT MATRIX IS UPDATED  AND FACTORIZED IN EACH ITERA-
C               TION USING  THE PIVOTAL STRATEGY  DETERMINED IN THE PRE-
C               VIOUS CALL TO THE SUBROUTINE
C             3-ADJOINT  MATRIX IS KEPT  CONSTANT FROM THE PREVIOUS CALL
C               TO THE SUBROUTINE
C     IFLAG   RETURN FLAG :
C             -3 INCORRECT PARAMETER IVT
C             -2 INCORRECT PARAMETER MODE
C             -1 WORKSPACE TO SMALL
C              0 NORMAL RETURN
C              1 LIMIT OF ITERATIONS REACHED
C              2 LIMIT OF TIME REACHED
C     OTPT    NUMBER OF OUTPUT UNIT
C
      INTEGER JRYT(1),ICYT(1),BTYP(1),OTPT
      REAL W(1),BCV(1)
      COMPLEX YT(1),V(1),Z
      COMMON /MDLFTTM/ JT,JICN,JICT,JIRT,JIKEEP,JIW,JCS,JVD,JRHS,JSENS,J
     1DEL,JICV,JCCV,JMAX
      IFLAG=0
      IT=0
      IF (IWRITE.LT.1) GO TO 10
      WRITE (OTPT,140) NB
   10 CALL SECOND (TS)
      TT=TS
```

```
      TA=0.
      IF (MODE.LE.3.OR.MODE.GE.1) GO TO 20
      IFLAG=-2
      RETURN
   20 IF (MODE.EQ.3) GO TO 40
      IF (IVT.LE.5.OR.IVT.GE.1) GO TO 30
      IFLAG=-3
      RETURN
   30 IF (MODE.EQ.2) GO TO 40
      NYT=NB+2*NTL
      N1=NB+NB
      NR=N1-2
      NZ=(7*NYT)/2
      LICN=3*NZ
      LIRN=NZ+NR
      JT=1
      JICN=JT+LICN
      JICT=JICN+LICN
      JIRT=JICT+NZ
      JIKEEP=JIRT+LIRN
      JIW=JIKEEP+5*NR
      JCS=JIW+8*NR
      JVD=JCS+N1
      JRHS=JVD+NR
      JSENS=JRHS+N1
      JDEL=JSENS+NR
      JICV=JDEL+NR
      JCCV=JICV+NR
      JMAX=JCCV+NR
   40 IF (JMAX.LE.LW) GO TO 50
      IFLAG=-1
      RETURN
   50 IF (MODE.EQ.2.OR.MODE.EQ.3) GO TO 60
      CALL DCVARF (W(JCCV),W(JICV),BTYP,NB-1)
   60 CALL STEP (YT,JRYT,ICYT,V,W(JVD),BTYP,BCV,W(JCS),W(JDEL),W(JT),W(J
     1IRT),W(JICT),W(JICN),W(JRHS),W(JIKEEP),W(JIW),W(JCCV),W(JICV),W(JS
     2ENS),NB,IVT,MODE,IP,EPS,OTPT,IWRITE)
      IT=IT+1
      IF (MODE.EQ.1) MODE=2
      CALL SECOND (TIME)
      TT=TIME-TS
      TI=TT-TA
      TA=TT
      IF (IWRITE.GE.1) WRITE (OTPT,150) IT,EPS,TI,TT
      IF (EPS.LE.VEPS) GO TO 90
      IF (IT.GE.ITEL) GO TO 70
      IF (TT.GE.TIMEL) GO TO 80
      GO TO 60
   70 IFLAG=1
      GO TO 90
   80 IFLAG=2
   90 VEPS=EPS
      TIMEL=TT
      ITEL=IT
      IF (IWRITE.LT.1) GO TO 130
      WRITE (OTPT,160) ITEL,IFLAG,VEPS,TIMEL
      WRITE (OTPT,170)
      JV1=JCS-1
      JV2=JSENS-1
      IF (IP.EQ.1) GO TO 110
      DO 100 I=1,NB
      Z=V(I)
      R1=CABS(Z)
      R2=ATAN2(AIMAG(Z),REAL(Z))
      W(JV1+I)=R1
```

```
      W(JV2+I)=R2
      WRITE (OTPT,180) I,Z,R1,R2
  100 CONTINUE
      GO TO 130
  110 DO 120 I=1,NB
      Z=V(I)
      R1=REAL(Z)
      R2=AIMAG(Z)
      R3=R1*COS(R2)
      R4=R1*SIN(R2)
      W(JV1+I)=R3
      W(JV2+I)=R4
      WRITE (OTPT,180) I,R3,R4,Z
  120 CONTINUE
  130 RETURN
C
  140 FORMAT (///" LOAD FLOW SOLUTION OF ",I3,"-BUS SYSTEM USING THE TEL
     1LEGEN THEOREM METHOD")
  150 FORMAT (//" IT = ",I2,3X,"EPS =",E10.4,3X,"ITERATION TIME ",F5.3,3
     1X,"TOTAL TIME ",F6.3/)
  160 FORMAT (///" RESULTS OF ANALYSIS"//" NUMBER OF ITERATIONS:",6X,I3/
     1" RETURN FLAG:",16X,I2/" ACCURACY OBTAINED:",2X,E10.4/" ANALYSIS T
     2IME:",10X,F6.3," SECONDS")
  170 FORMAT (/" VECTOR OF BUS VOLTAGES"/4X,"BUS",7X,"RECTANGULAR COORDI
     1NATES",10X,"POLAR COORDINATES"/)
  180 FORMAT (3X,I4,2(3X,2(1X,E13.7)))
      END
```

```
C
C
C
      SUBROUTINE MISM (BCV,BCS,V,BTYP,UDEL,N,IWRITE)
C
C     11.05.1982.
C
C     SUBROUTINE MISM CALCULATES REAL VECTOR UDEL OF MISMATCHES
C     OF BUS CONTROL VARIABLES
C
      INTEGER BTYP(1)
      REAL BCV(1),UDEL(1)
      COMPLEX BCS(1),V(1),Y,VA
      DO 30 I=1,N
      J1=2*I-1
      J2=2*I
      K=BTYP(I)
      IF (K.NE.0) GO TO 10
C
C     SETTING UP VALUES OF ELEMENTS OF UDEL  CORRESPONDING TO A LOAD BUS
C
      Y=BCS(I)
      UDEL(J1)=BCV(J1)-REAL(Y)
      UDEL(J2)=BCV(J2)-AIMAG(Y)
      GO TO 30
   10 IF (K.NE.1) GO TO 20
C
C     SETTING UP VALUES OF ELEMENTS OF UDEL CORRESPONDING TO A GENERATOR
C
      UDEL(J1)=BCV(J1)-REAL(BCS(I))
      UDEL(J2)=BCV(J2)-CABS(V(I))
      GO TO 30
C
C     SETTING UP VALUES OF ELEMENTS OF  UDEL  CORRESPONDING TO SLACK BUS
C
   20 VA=V(I)
      UDEL(J1)=BCV(J1)-CABS(VA)
      UDEL(J2)=BCV(J2)-ATAN2(AIMAG(VA),REAL(VA))
   30 CONTINUE
      IF (IWRITE.LT.4) GO TO 40
      WRITE (6,50)
      WRITE (6,60) (I,UDEL(2*I-1),UDEL(2*I),I=1,N)
   40 RETURN
   50 FORMAT (//" VECTOR UDEL OF MISMATCHES OF BUS CONTROL VARIABLES"/"
     1VALUE OF AN ELEMENT IS PRECEDED BY THE NUMBER OF BUS"/)
   60 FORMAT (3(2X,I4,":",2(1X,E13.7)))
      END
```

```
C
C
C
      SUBROUTINE PQ (V,YT,JRYT,ICYT,BCS,N,IWRITE)
C
C     06.05.1982.
C
C     SUBROUTINE PQ CALCULATES VECTOR BCS OF COMPLEX POWERS INJECTED
C     INTO BUSES FOR THE GIVEN VECTOR V OF BUS VOLTAGES
C
      COMPLEX V(1),YT(1),BCS(1)
      INTEGER JRYT(1),ICYT(1)
      DO 10 I=1,N
      BCS(I)=(0.,0.)
   10 CONTINUE
      DO 30 I=1,N
      J1=JRYT(I)
      J2=JRYT(I+1)-1
      DO 20 J=J1,J2
      BCS(I)=BCS(I)+YT(J)*V(ICYT(J))
   20 CONTINUE
      BCS(I)=V(I)*CONJG(BCS(I))
   30 CONTINUE
      IF (IWRITE.LT.3) RETURN
      WRITE (6,40)
      WRITE (6,50) (I,BCS(I),I=1,N)
      RETURN
   40 FORMAT (//" VECTOR BCS OF COMPLEX POWERS INJECTED INTO BUSES"/" VA
     1LUE OF AN ELEMENT IS PRECEDED BY THE NUMBER OF BUS "/)
   50 FORMAT (3(3X,I3,":",2(1X,E13.7)))
      END
```

```
C
C
C
      SUBROUTINE PTELT (T,NT,R,IRT,ICT,IR,IC)
C
C     06.05.1982.
C
C     SUBROUTINE PTELT PLACES A NONZERO ELEMENT OF THE ADJOINT
C     MATRIX  INTO THE VECTOR T AND  STORES ITS ROW AND COLUMN
C     INDICES IN VECTORS IRT AND ICT
C
      INTEGER IRT(1),ICT(1)
      REAL T(1)
      NT=NT+1
      T(NT)=R
      IRT(NT)=IR
      ICT(NT)=IC
      RETURN
      END
```

```
C
C
C
      SUBROUTINE RDAT (LBINP,LBOUT,LINPG,LINPB,LG,LB,LOUTG,LOUTB,LTAP,BN
     1R,BTYP,BVMOD,BVARG,BGP,BLP,BLQ,BSTL,JRYT,NB,NTL,NLB,INPT,IWRITE)
C
C     05.05.1982.
C
C     SUBROUTINE RDAT READS INPUT DATA DESCRIBING POWER SYSTEM FROM
C     A FILE CREATED BY SUBROUTINE FORMDTF AND PREPROCESS THIS DATA
C
      INTEGER LBINP(1),LBOUT(1),BNR(1),BTYP(1),JRYT(1)
      REAL LINPG(1),LINPB(1),LG(1),LB(1),LOUTG(1),LOUTB(1),LTAP(1),BVMOD
     1(1),BVARG(1),BGP(1),BLP(1),BLQ(1),BSTL(1),HDLN(12)
      COMPLEX Z
      READ (INPT,110) (HDLN(I),I=1,12)
      READ (INPT,130) X,NB,X,X,NTL
C
C     BECAUSE BUS IS INCIDENT TO ITSELF THEN ALL ELEMENTS OF VECTOR
C     JRYT ARE PRESET TO ONE
C
      DO 10 I=1,NB
      JRYT(I)=1
   10 CONTINUE
      IF (IWRITE.LT.1) GO TO 20
      WRITE (6,110) (HDLN(I),I=1,12)
      WRITE (6,150) NB,NTL
   20 READ (INPT,110) (HDLN(I),I=1,12)
      IF (IWRITE.LT.2) GO TO 30
      WRITE (6,110) (HDLN(I),I=1,12)
   30 READ (INPT,120) (HDLN(I),I=1,12)
      DO 40 I=1,NTL
      READ (INPT,140) LBINP(I),LBOUT(I),LINPG(I),LINPB(I),LG(I),LB(I),LO
     1UTG(I),LOUTB(I),LTAP(I)
C
C     DEGREES OF BOTH BUSES INCIDENT TO THE LINE ARE INCREASED
C     BY ONE
C
      J=LBINP(I)
      JRYT(J)=JRYT(J)+1
      J=LBOUT(I)
      JRYT(J)=JRYT(J)+1
   40 CONTINUE
      IF (IWRITE.LT.2) GO TO 60
      WRITE (6,120) (HDLN(I),I=1,12)
      DO 50 I=1,NTL
      WRITE (6,140) LBINP(I),LBOUT(I),LINPG(I),LINPB(I),LG(I),LB(I),LOUT
     1G(I),LOUTB(I),LTAP(I)
   50 CONTINUE
   60 DO 70 I=1,NTL
      Z=1./CMPLX(LG(I),LB(I))
      LG(I)=REAL(Z)
      LB(I)=AIMAG(Z)
   70 CONTINUE
      READ (INPT,110) (HDLN(I),I=1,12)
      IF (IWRITE.GT.1) WRITE (6,110) (HDLN(I),I=1,12)
      READ (INPT,120) (HDLN(I),I=1,12)
      NLB=0
      IX=JRYT(1)
      JRYT(1)=1
      JRYT(NB+1)=0
      DO 80 I=1,NB
      READ (INPT,140) L,BTYP(L),BVMOD(L),BVARG(L),BGP(L),BLP(L),BLQ(L),B
     1STL(L)
      BNR(L)=I
```

```
      IF (BTYP(L).EQ.0) NLB=NLB+1
      J=I+1
      IY=JRYT(J)
      JRYT(J)=JRYT(I)+IX
      IX=IY
   80 CONTINUE
      IF (IWRITE.LT.2) GO TO 100
      WRITE (6,120) (HDLN(I),I=1,12)
      DO 90 I=1,NB
      WRITE (6,140) BNR(I),BTYP(I),BVMOD(I),BVARG(I),BGP(I),BLP(I),BLQ(I
     1),BSTL(I)
   90 CONTINUE
  100 RETURN
  110 FORMAT (//12A10/)
  120 FORMAT (12A10)
  130 FORMAT (1X,A5,I3,2A5,I4)
  140 FORMAT (2(1X,I5),2X,7(2X,E13.7))
  150 FORMAT (" NB = ",I3.3,",      ","NTL =",I4.3)
      END
```

```
C
C
C
      SUBROUTINE READDT (LBINP,LBOUT,LINPG,LINPB,LR,LX,LOUTG,LOUTB,LTAP,
     1BNR,BTYP,BVMOD,BVARG,BGP,BLP,BLQ,BSTL,NB,NTL,INPT)
C
C     27.04.1982.
C
C     SUBROUTINE READDT READS INPUT DATA DESCRIBING THE POWER SYSTEM
C     FROM AN UMFORMATTED FILE WITH THE USE OF PACKAGE PWRDD
C
C     LIBRARY : LIBSPWR (PACKAGE PWRDD)
C
      INTEGER LBINP(1),LBOUT(1),BNR(1),BTYP(1),KK(2)
      REAL LINPG(1),LINPB(1),LR(1),LX(1),LOUTG(1),LOUTB(1),LTAP(1),BVMOD
     1(1),BVARG(1),BGP(1),BLP(1),BLQ(1),BSTL(1),RLB(17),RL(9),RB(8)
      EQUIVALENCE (RLB(1),RL(1)), (RLB(10),RB(1))
      DATA RLB/"LINEBINP","LINEBOUT","LINEINPC","LINEINPS","LINER","LINE
     1X","LINEOUTC","LINEOUTS","LINETAP","BUSNR","BUSTYPE","BUSVMOD","BU
     2SVARG","BUSGP","BUSLP","BUSLQ","BUSSTL"/
      R=ATAN(1.)/45.
      CALL DD00DF (INPT,NB,NS,NTL,TN,ICODE)
      IF (ICODE.LT.0) GO TO 70
      CALL DD00DN (RLB,17,IE)
      IF (IE.NE.0) GO TO 80
      CALL DD11DR (1,RL,9,IE)
      IF (IE.NE.0) GO TO 90
      CALL DD11DR (2,RB,8,IE)
      IF (IE.NE.0) GO TO 100
      DO 10 I=1,2
      ICODE=ICODE/10
      KK(I)=MOD(ICODE,10)
   10 CONTINUE
      I=1
   20 CALL DD11GS (1,IE)
      IF (IE.EQ.0) GO TO 30
      CALL DD11IN (LBINP(I),IE)
      CALL DD11IN (LBOUT(I),IE)
      CALL DD11RN (LINPG(I),IE)
      CALL DD11RN (LINPB(I),IE)
      CALL DD11RN (LR(I),IE)
      CALL DD11RN (LX(I),IE)
      CALL DD11RN (LOUTG(I),IE)
      CALL DD11RN (LOUTB(I),IE)
      CALL DD11RN (LTAP(I),IE)
      I=I+1
      GO TO 20
   30 IF (KK(2).LT.KK(1)) GO TO 40
      REWIND INPT
   40 I=1
   50 CALL DD11GS (2,IE)
      IF (IE.EQ.0) GO TO 60
      CALL DD11IN (BNR(I),IE)
      CALL DD11IN (BTYP(I),IE)
      CALL DD11RN (BVMOD(I),IE)
      CALL DD11RN (ALPHA,IE)
      BVARG(I)=ALPHA*R
      CALL DD11RN (BGP(I),IE)
      CALL DD11RN (BLP(I),IE)
      CALL DD11RN (BLQ(I),IE)
      CALL DD11RN (BSTL(I),IE)
      I=I+1
      GO TO 50
   60 RETURN
   70 JD=1
```

```
      IE=ICODE
      GO TO 110
   80 JD=2
      GO TO 110
   90 JD=3
      GO TO 110
  100 JD=4
  110 WRITE (6,120) JD,IE
      STOP
  120 FORMAT (///1X,"INCORRECT INPUT DATA. STOP",I2,". IFLAG =",I2/)
      END
```

```
C
C
C
      SUBROUTINE SENSIT (LBINP,LBOUT,YT,JRYT,ICYT,V,BTYP,BCS,RHST,CCV,IC
     1V,PDR,SENS,T,IRT,ICT,ICN,IKEEP,IW,NB,NCV,IVT,MODE,IFLAG,OTPT,IWRIT
     2E)
C
C     12.08.1982.
C
C     SUBROUTINE  SENSIT IS THE HIGHEST  LEVEL SUBROUTINE TO CALCULATE
C     VECTOR  OF SENSITIVITIES OF A REAL FUNCTION  OF THE POWER SYSTEM
C     STATE AND CONTROL  VARIABLES W.R.T. CONTROL VARIABLES DEFINED BY
C     VECTORS CCV,ICV USING A SPARSE MATRIX TECHNIQUE (HARWELL PACKAGE
C     MA28) AND THE TELLEGEN THEOREM METHOD
C
C     LIBRARY : LIBRHSM (HARWELL PACKAGE MA28)
C
      INTEGER LBINP(1),LBOUT(1),JRYT(1),ICYT(1),BTYP(1),CCV(1),ICV(1),IR
     1T(1),ICT(1),ICN(1),IKEEP(1),IW(1),OTPT
      REAL RHST(1),T(1),SENS(1),HD(5)
      COMPLEX YT(1),V(1),BCS(1),PDR(1)
      COMMON /T00/ NT,LICN,LIRN
      DATA HD/"EXACT","DECOUPLED","APPROXIM","APPR DEC","MIXED"/
      IFLAG=0
      N=NB-1
      NR=N+N
      MTYPE=1
      IF (IWRITE.LT.3) GO TO 10
      WRITE (OTPT,130) MODE,HD(IVT)
   10 IF (MODE.LE.3.OR.MODE.GE.1) GO TO 20
      IFLAG=-2
      RETURN
   20 IF (MODE.EQ.3) GO TO 120
C
C     IF MODE=3 THEN MATRIX T OF ADJOINT EQUATIONS IS NOT CHANGED
C     STRUCTURALLY  AND  NUMERICALLY  FROM  THE PREVIUOS  CALL TO
C     SUBROUTINE SENSIT, STEP OR LFTTM
C
      IF (IVT.LE.5.OR.IVT.GE.1) GO TO 30
      IFLAG=-3
      RETURN
   30 IF (IVT.LE.2.OR.IVT.EQ.5) CALL PQ (V,YT,JRYT,ICYT,BCS,NB,IWRITE-1)
      GO TO (40,50,60,70,80), IVT
C
C     PARAMETER IVT IS TO SELECT VERSION OF ADJOINT EQUATIONS
C
   40 CALL FORMTE (YT,JRYT,ICYT,BCS,V,BTYP,T,IRT,ICT,N,NT,OTPT,IWRITE)
      GO TO 90
   50 CALL FORMTD (YT,JRYT,ICYT,BCS,V,BTYP,T,IRT,ICT,N,NT,OTPT,IWRITE)
      GO TO 90
   60 CALL FORMTA (YT,JRYT,ICYT,BTYP,T,IRT,ICT,N,NT,OTPT,IWRITE)
      GO TO 90
   70 CALL FORMTAD (YT,JRYT,ICYT,BTYP,T,IRT,ICT,N,NT,OTPT,IWRITE)
      GO TO 90
   80 CALL FORMTM (YT,JRYT,ICYT,BCS,V,BTYP,T,IRT,ICT,N,NT,OTPT,IWRITE)
   90 IF (MODE.EQ.2) GO TO 110
C
C     IF MODE=2 THEN MATRIX T OF ADJOINT EQUATIONS IS NOT CHANGED
C     STRUCTURALLY FROM THE PREVIOUS  CALL TO  SUBROUTINE SENSIT,
C     STEP OR LFTTM
C
      LICN=3*NT
      LIRN=NR+NT
      U=0.1
      DO 100 I=1,NT
```

```
      ICN(I)=ICT(I)
  100 CONTINUE
      CALL MA28A (NR,NT,T,LICN,IRT,LIRN,ICN,U,IKEEP,IW,BCS,IFLAG)
      GO TO 120
  110 CALL MA28B (NR,NT,T,LICN,IRT,ICT,ICN,IKEEP,IW,BCS,IFLAG)
  120 CALL MA28C (NR,T,LICN,ICN,IKEEP,RHST,BCS,MTYPE)
      CALL DERIV (LBINP,LBOUT,YT,JRYT,ICYT,V,RHST,CCV,ICV,PDR,SENS,NCV,I
     1WRITE)
      RETURN
  130 FORMAT (///" SENSITITVITY CALCULATION. MODE =",I2/" VERSION OF ADJ
     1OINT EQUATIONS: ",A9)
      END
```

```
C
C
C
      SUBROUTINE STEP (YT,JRYT,ICYT,V,VD,BTYP,BCV,BCS,UDEL,T,IRT,ICT,ICN
     1,RHST,IKEEP,IW,CCVF,ICVF,SENS,NB,IVT,MODE,IP,EPS,OTPT,IWRITE)
C
C     03.08.1982.
C
C     SUBROUTINE STEP PERFORMS ONE ITERATION OF THE SOLUTION OF
C     THE LOAD FLOW PROBLEM  USING THE TELLEGEN  THEOREM METHOD
C
C     LIBRARY : LIBRHSM (HARWELL PACKAGE MA28)
C
      INTEGER JRYT(1),ICYT(1),BTYP(1),IRT(1),ICT(1),ICN(1),IKEEP(1),IW(1
     1),CCVF(1),ICVF(1),LBINP(1),LBOUT(1),OTPT
      REAL HD(7),BCV(1),UDEL(1),T(1),RHST(1),SENS(1)
      COMPLEX YT(1),V(1),VD(1),BCS(1),Z,Y,CRP
      COMMON /T00/ NT,LICN,LIRN
      DATA HD/"EXACT","DECOUPLED","APPROXIM","APPR DEC","MIXED","RECTANG
     1","POLAR"/
C
C     THIS FUNCTION STATEMENT IS TO CONVERSE COMPLEX NUMBERS FROM
C     RECTANGULAR INTO POLAR COORDINATES
C
      CRP(Y)=CMPLX(CABS(Y),ATAN2(AIMAG(Y),REAL(Y)))
      N=NB-1
      NR=2*N
      MTYPE=1
      EPS=0.
      RHST(NR+1)=0.
      RHST(NR+2)=0.
      IF (IWRITE.LT.2) GO TO 10
      I=6
      IF (IP.EQ.1) I=7
      WRITE (OTPT,240) MODE,HD(I),HD(IVT)
   10 IF (IP.NE.1) GO TO 30
C
C     VECTOR V OF INITIAL BUS VOLTAGES GIVEN IN POLAR COORDINATES
C     IS CONVERSED INTO RECTANGULAR COORDINATES
C
      DO 20 I=1,NB
      R1=REAL(V(I))
      R2=AIMAG(V(I))
      V(I)=CMPLX(R1*COS(R2),R1*SIN(R2))
   20 CONTINUE
   30 CALL PQ (V,YT,JRYT,ICYT,BCS,N,IWRITE)
      CALL MISM (BCV,BCS,V,BTYP,UDEL,N,IWRITE)
      IF (MODE.EQ.3) GO TO 120
C
C     IF MODE=3 THEN MATRIX T OF ADJOINT EQUATIONS IS NOT CHANGED
C     STRUCTURALLY  AND  NUMERICALLY  FROM  THE PREVIUOS  CALL TO
C     SUBROUTINE STEP, LFTTM OR SENSIT
C
      GO TO (40,50,60,70,30), IVT
C
C     PARAMETER IVT IS TO SELECT VERSION OF ADJOINT EQUATIONS
C
   40 CALL FORMTE (YT,JRYT,ICYT,BCS,V,BTYP,T,IRT,ICT,N,NT,OTPT,IWRITE)
      GO TO 90
   50 CALL FORMTD (YT,JRYT,ICYT,BCS,V,BTYP,T,IRT,ICT,N,NT,OTPT,IWRITE)
      GO TO 90
   60 CALL FORMTA (YT,JRYT,ICYT,BTYP,T,IRT,ICT,N,NT,OTPT,IWRITE)
      GO TO 90
   70 CALL FORMTAD (YT,JRYT,ICYT,BTYP,T,IRT,ICT,N,NT,OTPT,IWRITE)
      GO TO 90
```

```
   80 CALL FORMTM (YT,JRYT,ICYT,BCS,V,BTYP,T,IRT,ICT,N,NT,OTPT,IWRITE)
   90 IF (MODE.EQ.2) GO TO 110
C
C     IF MODE=2 THEN MATRIX T OF ADJOINT EQUATIONS IS NOT CHANGED
C     STRUCTURALLY  FROM  THE PREVIOUS  CALL TO SUBROUTINE  STEP,
C     LFTTM OR SENSIT
C
      LICN=3*NT
      LIRN=NR+NT
      U=0.1
      DO 100 I=1,NT
      ICN(I)=ICT(I)
  100 CONTINUE
      CALL MA28A (NR,NT,T,LICN,IRT,LIRN,ICN,U,IKEEP,IW,SENS,IFLAG)
      GO TO 120
  110 CALL MA28B (NR,NT,T,LICN,IRT,ICT,ICN,IKEEP,IW,SENS,IFLAG)
  120 DO 210 I=1,NR
C
C     I IS THE INDEX OF STATE VARIABLE. IN THIS LOOP CORRECTIONS
C     OF ALL STATE VARIABLES ARE CALCULATED
C
      COR=0.
      K=MOD(I,2)
      L=(I+K)/2
C
C     L IS THE INDEX OF BUS ASSOCIATED WITH THE ITH STATE VARIABLE
C     IF ITH STATE  VARIABLE  IS REAL(V(L)) OR MOD(V(L))   THEN K=1
C     IF ITH STATE  VARIABLE  IS AIMAG(V(L)) OR ARG(V(L)) THEN K=0
C
      ITYP=BTYP(L)
      IF (ITYP.NE.1.OR.K.NE.1.OR.IP.NE.1) GO TO 130
C
C     IF THE STATE VARIABLE IS A MODULUS OF GENERATOR VOLTAGE THEN
C     ADJOINT EQUATIONS ARE HOMOGENEOUS AND NEED NOT BE SOLVED
C
      Z=CMPLX(0.,0.)
      GO TO 210
  130 CALL EXCT (V,ITYP,I,RHST,NR,IP,IWRITE)
      CALL MA28C (NR,T,LICN,ICN,IKEEP,RHST,SENS,MTYPE)
      CALL DERIV (LBINP,LBOUT,YT,JRYT,ICYT,V,RHST,CCVF,ICVF,PDR,SENS,NR,
     1IWRITE)
C
C     CALCULATION OF THE VALUE OF CORRECTION COR OF THE ITH STATE
C     VARIABLE
C
      IF (IP.EQ.1) GO TO 150
C
C     RECTANGULAR VERSION OF THE POWER FLOW EQUATIONS
C
      DO 140 J=1,NR
      R=0.
C
C     VARIABLE R HOLDS THE VALUE OF PARTIAL DERIVATIVE OF ITH STATE
C     VARIABLE W.R.T. JTH CONTROL VARIABLE
C
      IF (CCVF(J).EQ.9.AND.I.EQ.(J-1)) R=REAL(V(L))/CABS(V(L))
      IF (CCVF(J).EQ.9.AND.I.EQ.J) R=AIMAG(V(L))/CABS(V(L))
      COR=COR+(R-SENS(J))*UDEL(J)
  140 CONTINUE
      GO TO 170
C
C     POLAR VERSION OF THE POWER FLOW EQUATIONS
C
  150 DO 160 J=1,NR
      COR=COR-SENS(J)*UDEL(J)
```

```
  160 CONTINUE
  170 IF (K.EQ.0) GO TO 180
      Z=CMPLX(COR,0.)
      GO TO 210
  180 Z=Z+CMPLX(0.,COR)
C
C     VARIABLE Z HOLDS THE VALUE OF VOLTAGE CORRECTION OF THE LTH BUS
C
      IF (IP.EQ.1) GO TO 190
      X=CABS(Z)
      VD(L)=V(L)+Z
      GO TO 200
  190 VD(L)=CRP(V(L))+Z
      R1=REAL(VD(L))
      R2=AIMAG(VD(L))
      X=CABS(CMPLX(R1*COS(R2),R1*SIN(R2))-V(L))
C
C     VARIABLE X HOLDS THE VALUE OF THE MODULUS OF THE BUS VOLTAGE
C     CORRECTION. VD(L) IS THE UPDATED VOLTAGE OF THE LTH BUS
C
  200 IF (X.GT.EPS) EPS=X
  210 CONTINUE
      DO 220 I=1,N
      V(I)=VD(I)
  220 CONTINUE
      IF (IP.EQ.1) V(NB)=CRP(V(NB))
      IF (IWRITE.LT.2) GO TO 230
      WRITE (OTPT,250)
      WRITE (OTPT,260) (L,V(L),L=1,N)
  230 RETURN
  240 FORMAT (///" RESULTS OF ITERATION. MODE =",I2/" VERSION OF POWER F
     1LOW EQUATIONS: ",A9/" VERSION OF ADJOINT EQUATIONS: ",A9)
  250 FORMAT (//" VECTOR V OF BUS VOLTAGES"/" VALUE OF VOLTAGE IS PRECED
     1ED BY THE NUMBER OF BUS"/)
  260 FORMAT (3(2X,I4,":",2(1X,E13.7)))
      END
```